

# Chapter 18

## Linear regression models

### 18.1 Crime and punishment

The data in `linearRegression_crimePunishment.csv` contains the murder rate per capita and the rate of automobile crimes per 100,000 individuals (both on the log scale) in the ten US States that have changed their legislation on capital punishment since 1960 (in all cases the states abolished capital punishment). We also include a dummy variable (“law”) that is 1 if the state allows capital punishment in that year, and 0 otherwise. The crime data is from <http://www.disastercenter.com>.

**Problem 18.1.1.** Graph the data and comment on any trends.

The data is shown in Figures 18.1 and 18.2. There seems to be some association between the murder rate and auto crimes. In all cases it is difficult to visually discern an impact of the change in legislation.

**Problem 18.1.2.** A simple model for murder rates is of the form,

$$murder_{i,t} \sim \mathcal{N}(\alpha + \beta penalty_{i,t} + \gamma car_{i,t}, \sigma) \quad (18.1)$$

where we assume that the effect of having the death penalty is given by  $\beta$ , which is assumed to be the same across all states. We include  $car_{i,t}$  – a measure of crimes on automobiles, as an independent variable to proxy for the contemporaneous underlying level of crime. Estimate this model and hence determine whether the death penalty acts as a deterrent to murder.

This model can be estimated using the following Stan code:

```
data{
  int N;
  int K;
  real murder[N];
  real car[N];
  int<lower=0, upper=1> law[N];
}
```




Figure 18.1: The log per capita murder rate (red) and auto crime rate (divided by 2 on log scale; green) versus the death penalty dummy (1 if state has death penalty; 0 otherwise; blue) in five of the US States which abolished the death penalty since 1960.

```

int state[N];
}

parameters{
  real alpha;
  real beta;
  real gamma;
  real<lower=0> sigma;
}

model{
  for(i in 1:N)
    murder[i] ~ normal(alpha + beta * law[i] + gamma * car[i], sigma);

  alpha ~ normal(0, 1);
  beta ~ normal(0, 1);
  gamma ~ normal(0, 1);
  sigma ~ normal(0, 1);
}

```




Figure 18.2: The log per capita murder rate (red) and auto crime rate (divided by 2 on log scale; green) versus the death penalty dummy (1 if state has death penalty; 0 otherwise; blue) in five of the US States which abolished the death penalty since 1960.

which when we run we obtain the following results,

```
## Print summary statistics
print(fit, probs = c(0.25, 0.5, 0.75))
```

	mean	se_mean	sd	2.5%	25%	50%	75%	97.5%	n_eff	Rhat
alpha	-1.48	0.02	0.19	-1.83	-1.62	-1.48	-1.35	-1.07	105	1.01
beta	0.24	0.00	0.06	0.10	0.20	0.24	0.27	0.34	219	1.00
gamma	0.45	0.00	0.04	0.38	0.43	0.45	0.48	0.52	105	1.01
sigma	0.62	0.00	0.02	0.59	0.61	0.62	0.64	0.66	345	0.99

Where we see that we estimate that the imposition of the death penalty on average *raises* the murder rate by on average 24%!

**Problem 18.1.3.** An alternative model allows there to be state-level effects,

$$\text{murder}_{i,t} \sim \mathcal{N}(\alpha_i + \beta_i \text{penalty}_{i,t} + \gamma_i \text{car}_{i,t}, \sigma_i), \quad (18.2)$$

where we assume that  $\alpha_i \sim \mathcal{N}(\bar{\alpha}, \sigma_\alpha)$ ,  $\beta_i \sim \mathcal{N}(\bar{\beta}, \sigma_\beta)$  and  $\gamma_i \sim \mathcal{N}(\bar{\gamma}, \sigma_\gamma)$  (we assume fully heterogeneous estimates for  $\sigma$ ). Estimate the above model and compare the results with the homogeneous

coefficient model.

This model can be estimated using the following code,

```
data{
  int N;
  int K;
  real murder[N];
  real car[N];
  int<lower=0, upper=1> law[N];
  int state[N];
}

parameters{
  real alpha[K];
  real beta[K];
  real gamma[K];
  real<lower=0> sigma[K];
  real alpha_top;
  real<lower=0> alpha_sigma;
  real beta_top;
  real<lower=0> beta_sigma;
  real gamma_top;
  real<lower=0> gamma_sigma;
}

model{
  for(i in 1:N)
    murder[i] ~ normal(alpha[state[i]] + beta[state[i]] * law[i]
      + gamma[state[i]] * car[i], sigma[state[i]]);

  alpha ~ normal(alpha_top, alpha_sigma);
  beta ~ normal(beta_top, beta_sigma);
  gamma ~ normal(gamma_top, gamma_sigma);
  alpha_top ~ normal(0, 1);
  beta_top ~ normal(0, 1);
  gamma_top ~ normal(0, 1);
  alpha_sigma ~ normal(0, 1);
  beta_sigma ~ normal(0, 1);
  gamma_sigma ~ normal(0, 1);
  sigma ~ normal(0, 1);
}

generated quantities{
  real alpha_average;
  real beta_average;
  real gamma_average;
```

```

alpha_average = normal_rng(alpha_top, alpha_sigma);
beta_average = normal_rng(beta_top, beta_sigma);
gamma_average = normal_rng(gamma_top, gamma_sigma);
}

```

which when we estimate the above yields,

```

## Print summary statistics
print(fit, probs = c(0.25, 0.5, 0.75))

```

mean	se_mean	sd	2.5%	25%	50%	75%	97.5%	n_eff	Rhat
alpha_average	-1.31	0.06	1.09	-3.27	-2.00	-1.37	-0.72	1.09	347 1.01
beta_average	-0.26	0.02	0.31	-0.90	-0.45	-0.27	-0.08	0.38	400 1.00
gamma_average	0.46	0.01	0.18	0.06	0.36	0.47	0.56	0.78	400 0.99

with a mean effect size of a 26% reduction in murder rates although with a much wider range of effects.

**Problem extra.** (Not in main text but wanted to include) Another model allows there to be time trends in the data,

$$\text{murder}_{i,t} \sim \mathcal{N}(\alpha_i + \delta_i t + \beta_i \text{penalty}_{i,t} + \gamma_i \text{car}_{i,t}, \sigma_i), \quad (18.3)$$

where  $\delta_i \sim \mathcal{N}(\bar{\delta}, \sigma_\delta)$ . Again estimate this model and compare the effect size of the death penalty across the three models.

This model is only a slight modification of the above one and yields an estimated effect of,

```

## Print summary statistics
print(fit, probs = c(0.25, 0.5, 0.75))

```

mean	se_mean	sd	2.5%	25%	50%	75%	97.5%	n_eff	Rhat
alpha_average	-1.76	0.06	1.04	-3.66	-2.44	-1.74	-1.07	0.30	348 1.00
beta_average	-0.18	0.01	0.26	-0.80	-0.32	-0.16	-0.04	0.35	331 1.00
gamma_average	0.54	0.01	0.15	0.19	0.45	0.55	0.63	0.84	400 1.01

And so an estimated reduction in the murder rate by 18% on average.

**Problem 18.1.4.** Compare the predictive fit of the models using the estimated leave-one-out cross-validation from the “loo” package. Which of the three models do you prefer? Hence conclude as to whether the death penalty acts as a deterrent of murder.

This can be done by recording the log likelihood of each data point in the `generated quantities` block. For example for the homogeneous coefficient model,

```

generated quantities{
  real logLikelihood[N];
  for(i in 1:N)
    logLikelihood[i] = normal_lpdf(murder[i] | alpha + beta * law[i] +
                                   gamma * car[i], sigma);
}

```

If we do this for all three models we obtain an estimated expected log (pointwise) likelihood from each using the `loo` function,

- Homogeneous: -512.2
- State-level, no time trend: -57.7
- State-level, with time trend: -3.9

Which when using the “compare” function we see that the best-fitting model is the State-level models with a time trend. In this model the estimated effect of the death penalty is to decrease the murder rate by 18% on average, with a 50% credible interval of  $4\% \leq effect \leq 32\%$ . So even though on average we see that there is quite a strong effect of the law, we are quite uncertain as to its size in a given state.

**Problem 18.1.5.** Critically evaluate the best-performing model and hence any conclusions that can be drawn from this analysis.

A criticism is that we have failed to include other omitted factors that may affect the murder rate but are also correlated with the abolishment of the death penalty. It is quite possible that contained within  $\alpha_i$  there are factors that may affect both the murder rate and be correlated with the redaction of the law. A potential improvement would be to use first-differences regression (or fixed effects),

$$\Delta murder_{i,t} \sim \mathcal{N}(\delta_i \Delta t + \beta_i \Delta penalty_{i,t} + \gamma_i \Delta car_{i,t}, \sigma_i), \quad (18.4)$$

where  $\Delta$  signifies the first difference of the variable. At least in the above model we have removed any individual (time-invariant) heterogeneity from affecting our estimates of the deterrent size. In fact, if we do estimate a model similar to the one above we do not see such any significant impact of the legislation (results not shown).

An alternative approach would be to include other factors in the model that explain  $\alpha_i$ .

# Bibliography